***********************************************************
/* 0. Program: 421-a Exemption */
***********************************************************

/*

Chetty, R., Hendren, N., & Katz, L. F. (2016). 
"The effects of exposure to better neighborhoods on children: New evidence from 
the Moving to Opportunity experiment."
American Economic Review, 106(4), 855-902.

Soltas, E. (2020).
"The Price of Inclusion: Evidence from Housing Developer Behavior."

*/

********************************
/* 1. Pull Global Assumptions */
********************************

*Program specific globals
local kids_age = "$kids_age" // "young", "old" or "observed"
local include_parents = "$include_parents" // "yes" or "no" to include parent impacts
local include_kids = "$include_kids" // "yes" or "no" to include kid impacts

local parent_earn_years = $parent_earn_years // either 2 or 10
local extend_parent_earn = "$extend_parent_earn" // "yes" or "no"
local years_enroll = $years_enroll
local ev_correction = "$ev_correction" // "yes" or "no"

*'global' globals
local discount_rate = 0.05 // assumption in rest of Soltas (2020)
local wage_growth_rate = $wage_growth_rate
local proj_age = $proj_age
local proj_type = "$proj_type" // "growth forecast"
local correlation = $correlation
local wtp_valuation = "$wtp_valuation"

*Tax Rate Globals
local tax_rate_assumption = "$tax_rate_assumption" 
local payroll_assumption = "$payroll_assumption" 

******************************
/* 2. Estimates from Paper */
******************************
/*
*Young
local earn_effect_young = 3476.8 //Chetty et al (2016) Table 3, Col. 4
local earn_effect_young_se = 1418.2 //Chetty et al (2016) Table 3, Col. 4
local tax_change_young = 393.6 //Chetty et al (2016) Table 12, Col. 3
local tax_change_young_se = 134.1 //Chetty et al (2016) Table 12, Col. 3
local college_effect_young = 5.233 // Chetty et al (2016) Appendix Table 4b, Col. 1
local college_effect_young_se = 2.382 // Chetty et al (2016) Appendix Table 4b, Col. 1

*Old
local earn_effect_old = -2426.7 //Chetty et al (2016) Table 3, Col. 4
local earn_effect_old_se = 2154.4 //Chetty et al (2016) Table 3, Col. 4
local tax_change_old = -441.6 //Chetty et al (2016) Table 12, Col. 3
local tax_change_old_se = 230.8 //Chetty et al (2016) Table 12, Col. 3
local college_effect_old = -10.32 // Chetty et al (2016) Appendix Table 4b, Col. 1
local college_effect_old_se = 4.221 // Chetty et al (2016) Appendix Table 4b, Col. 1


*Parent impacts
*Earnings
local earn_impact_1_2 = -786.86 // HUD MTO Final impact report exhibit 5.5, TOT impact experimental vs control
local earn_impact_1_2_se = 462.10  // HUD report exhibit 5.5, TOT impact experimental vs control
local earn_impact_1_10 = -672.54 // HUD report exhibit 5.5, TOT impact experimental vs control
local earn_impact_1_10_se = 716.54 // HUD report exhibit 5.5, TOT impact experimental vs control
*/

/* Import estimates from paper, giving option for corrected estimates.
When bootstrap!=yes import point estimates for causal estimates.
When bootstrap==yes import a particular draw for the causal estimates.
${folder_name}, being set externally, may vary in order to use pub bias corrected estimates. */

if "`1'" != "" global name = "mto"
local bootstrap = "`2'"
if "`3'" != "" global folder_name = "`3'"
if "`bootstrap'" == "yes" {
	if ${draw_number} ==1 {
		preserve
			use "${input_data}/causal_estimates/${folder_name}/draws/${name}.dta", clear
			qui ds draw_number, not 
			global estimates_${name} = r(varlist)
			
			mkmat ${estimates_${name}}, matrix(draws_${name}) rownames(draw_number)
		restore
	}
	local ests ${estimates_${name}}
	foreach var in `ests' {
		matrix temp = draws_${name}["${draw_number}", "`var'"]
		local `var' = temp[1,1]
	}
}
if "`bootstrap'" != "yes" {
	preserve
		import delimited "${input_data}/causal_estimates/${folder_name}/${name}.csv", clear
		levelsof estimate, local(estimates)
		foreach est in `estimates' {
			qui su pe if estimate == "`est'"
			local `est' = r(mean)
		}
	restore
}

*********************************
/* 3. Assumptions from Paper */
*********************************

*local earn_control_1_2 = 14050 // Estimate from Picture of Subsidized Households
*local earn_control_1_10 = 14050 // Estimate from Picture of Subsidized Households

local earn_control_1_2 = 37355 // Median of 30-60% AMI (top comes from 421-a income restriction, bottom from very low income band)
local earn_control_1_10 = 37355 // (lower limit exists b/c cannot be "rent burdened" ie 35% of income on rent)

* Use MTO control parent earnings for computation of % effects on parent earnings
local earn_control_1_2_mto = 5781.17 // HUD report exhibit 5.5 
local earn_control_1_10_mto = 9092.08 // HUD report exhibit 5.5

* Use MTO control child earnings for same purpose
local MTO_control_earn_26_young = 11398.3 //Chetty et al (2016) Table 3
local MTO_control_earn_26_old = 13968.9 //Chetty et al (2016) Table 3

local frac_kids_young = 0.608 // CPS tabulation using SPM match of housing aid recipients in NYC in 2003-2014

local avg_age_move_young = 6.041844 // CPS
local avg_age_move_old = 15.44037 // CPS

local young_household_head = 40.50477 // CPS
local old_household_head = 44.55792 // CPS

if  "`kids_age'" == "young" {
	local age_kid = `avg_age_move_young' // CPS
	local age_stat = `young_household_head' // CPS
}
if  "`kids_age'" == "old" {
	local age_kid = `avg_age_move_old' // CPS
	local age_stat = ``old_household_head'' // CPS
}
if  "`kids_age'" == "observed" {
	local age_kid = `frac_kids_young'*`avg_age_move_young'+ (1-`frac_kids_young')*`avg_age_move_old' // CPS
	local age_stat = `frac_kids_young'*`young_household_head' + (1-`frac_kids_young')*`old_household_head' // CPS
}

local usd_year = 2018

/*
Dataset of 421-a buildings issued initial construction permits from 2003 - 2014.
Use year of completion for average building, weighted by units, in PLUTO
*/
local year_move = round(2010.592,1)
local year_move_mto = 1996

local mean_voucher_start_year = 2010.592
local mean_voucher_start_year_int = round(`mean_voucher_start_year',1)

local year_cost = $year_cost // Soltas estimate for citywide, less 11k for lihtc/section 8
local pct_diff = $pct_diff

get_wtp_housing $year_cost 
local year_val = r(wtp)
* ub: local year_val = $year_cost

*local ch_povrate_young_mto = -0.3596 //Chetty et al (2016) pg. 14
*local ch_povrate_old_mto = -0.3470 //Chetty et al (2016) pg. 14

local kids_per_family = 1.445012 // CPS

local benefit_ratio = 1-$incidence // use incidence estimate

if "`ev_correction'" == "yes" local ev_coeff = `benefit_ratio'
else local ev_coeff = 1

**********************************
/* 4. Intermediate Calculations */
**********************************

*Deflate parent impacts from HUD report
deflate_to 2012, from(2009) // HUD report is in 2009 dollars, Chetty et al. (2012)
local deflate_09_12 = r(deflator)
foreach local in earn_impact_1_2 earn_control_1_2_mto earn_impact_1_10 earn_control_1_10_mto {
	local `local' = ``local''*`deflate_09_12'
}

*Rescale parent impacts from MTO to mean parent incomes of 421-a HHs
local earn_impact_1_2 = `earn_control_1_2 ' * `earn_impact_1_2' / `earn_control_1_2_mto'
local earn_impact_1_10 = `earn_control_1_10 ' * `earn_impact_1_10' / `earn_control_1_10_mto'

*Define child impacts from MTO in percentage terms, rescale by change in povrate
local earn_effect_young_pct = `pct_diff' 
local earn_effect_old_pct = `pct_diff'

local avg_age_move_young_int = round(`avg_age_move_young',1)
local avg_age_move_old_int = round(`avg_age_move_old',1)


*Estimating childrens' incomes
if "`proj_type'" == "growth forecast" {

	di "`proj_type'"
	di `earn_effect_young_pct'
	di `year_move'
	di `avg_age_move_young_int'
	di `usd_year'
	di `earn_control_1_10'
	di "$earn_method"
	di `young_household_head'
	
	est_life_impact_nyc `earn_effect_young_pct',  ///
		impact_age(26) project_age(18) end_project_age(26) ///
		project_year(`=`year_move'+18-`avg_age_move_young_int'') usd_year(`usd_year') ///
		income_info(`earn_control_1_10') income_info_type(parent_income) ///
		earn_method($earn_method) tax_method(off) transfer_method(off) ///
		parent_age(`=round(`young_household_head',1)') ///
		max_age_obs(26) parent_income_year(2015) percentage(yes)

	local cfactual_income_young = r(cfactual_income) 
	local tot_earn_impact_dy = r(tot_earn_impact_d)
	local tot_earn_impact_y = `tot_earn_impact_dy'*((1/(1+`discount_rate'))^(18-`avg_age_move_young_int'))
	
	est_life_impact_nyc `earn_effect_young_pct', ///
		impact_age(26) project_age(27) end_project_age(`proj_age') ///
		project_year(`=`year_move'+18-`avg_age_move_young_int' + 9') usd_year(`usd_year') ///
		income_info(`earn_control_1_10') income_info_type(parent_income) ///
		earn_method($earn_method) tax_method(off) transfer_method(off) ///
		max_age_obs(26) parent_income_year(2015) percentage(yes)  
	
	local tot_earn_impact_y = `tot_earn_impact_y' + r(tot_earn_impact_d)*((1/(1+`discount_rate'))^(27-`avg_age_move_young_int'))
	
	if "`kids_age'" == "old" {
	
		local tot_earn_impact_y = 0 
		local tax_impact_y = 0
		local tot_earn_impact_aftertax_y = 0
		
	}
	
	if "`kids_age'" == "observed" | "`kids_age'" == "old"  {
		
		est_life_impact_nyc `earn_effect_old_pct', ///
			impact_age(26) project_age(18) end_project_age(26) ///
			project_year(`=`year_move'+18-`avg_age_move_old_int'') usd_year(`usd_year') ///
			income_info(`earn_control_1_10') income_info_type(parent_income) ///
			earn_method($earn_method) tax_method(off) transfer_method(off) ///
			parent_age(`=round(`old_household_head',1)') ///
			max_age_obs(26) parent_income_year(2015) percentage(yes) 
		
		local cfactual_income_old = r(cfactual_income) 
		local tot_earn_impact_do = r(tot_earn_impact_d)
		local tot_earn_impact_o = r(tot_earn_impact_d)*((1/(1+`discount_rate'))^(18-`avg_age_move_old_int'))
			
		est_life_impact_nyc `earn_effect_old_pct', ///
			impact_age(26) project_age(27) end_project_age(`proj_age') ///
			project_year(`=`year_move'+18-`avg_age_move_old_int' + 9') usd_year(`usd_year') ///
			income_info(`earn_control_1_10') income_info_type(parent_income) ///
			earn_method($earn_method) tax_method(off) transfer_method(off) ///
			parent_age(`=round(`old_household_head',1)') ///
			max_age_obs(26) parent_income_year(2015) percentage(yes)  
		
		local tot_earn_impact_o = `tot_earn_impact_o' + r(tot_earn_impact_d)*((1/(1+`discount_rate'))^(27-`avg_age_move_old_int'))
			
	}
	
}

*Estimating tax rates 
if "`tax_rate_assumption'" == "paper internal" {
	local tax_rate_young_short = `tax_change_young'/`earn_effect_young'
	local tax_rate_old_short = `tax_change_old'/`earn_effect_old'
	local tax_rate_young_long = `tax_change_young'/`earn_effect_young'
	local tax_rate_old_long = `tax_change_old'/`earn_effect_old'
	local tax_rate_parent = $tax_rate_cont
}

if "`tax_rate_assumption'" == "continuous" {
	local tax_rate_young_short = $tax_rate_cont
	local tax_rate_old_short = $tax_rate_cont
	local tax_rate_young_long = $tax_rate_cont
	local tax_rate_old_long = $tax_rate_cont
	local tax_rate_parent = $tax_rate_cont
}

if "`tax_rate_assumption'" == "mixed" {

	local tax_rate_young_short = `tax_change_young'/`earn_effect_young'
	local tax_rate_old_short = `tax_change_old'/`earn_effect_old'
	
	*Kid tax rates
	foreach age in young old {		
		get_tax_rate_nyc `cfactual_income_`age'', ///
			 include_transfers(yes) ///
			 include_payroll(`payroll_assumption') /// "yes" or "no"
			 forecast_income(yes) /// forecast long-run earnings, so we get a realistic lifetime MTR.
			 usd_year(`usd_year') /// USD year of income
			 inc_year(`=round(`year_move'+26-`avg_age_move_`age'',1)') /// year of income measurement
			 earnings_type(individual) ///
			 program_age(26) // age we're projecting from
			 
		local tax_rate_`age'_long = r(tax_rate)
	}
	
	*Parent tax rate
	get_tax_rate_nyc `earn_control_1_`parent_earn_years'', ///
			 include_transfers(yes) ///
			 include_payroll(`payroll_assumption') /// "yes" or "no"
			 forecast_income(no) ///
			 usd_year(`usd_year') /// USD year of income
			 inc_year(`=`year_move'+(`parent_earn_years'/2)') /// year of income measurement
			 earnings_type(household) ///
			 program_age(`=round(`age_stat',1)') /// age we're projecting from
			 kids(`=round(`kids_per_family')')
			 
	local tax_rate_parent = r(tax_rate)
	
}

if "`tax_rate_assumption'" == "cbo" {

	*Kid tax rates
	foreach age in young old {		
		get_tax_rate_nyc `cfactual_income_`age'', ///
			 include_transfers(yes) ///
			 include_payroll(`payroll_assumption') /// "yes" or "no"
			 forecast_income(yes) /// forecast long-run earnings, so we get a realistic lifetime MTR.
			 usd_year(`usd_year') /// USD year of income
			 inc_year(`=round(`year_move'+26-`avg_age_move_`age'',1)') /// year of income measurement
			 earnings_type(individual) ///
			 program_age(26) // age we're projecting from
			 
		local tax_rate_`age'_short = r(tax_rate)
		local tax_rate_`age'_long = r(tax_rate)
	}
	
	*Parent tax rate
	get_tax_rate_nyc `earn_control_1_`parent_earn_years'', ///
			 include_transfers(yes) ///
			 include_payroll(`payroll_assumption') /// "yes" or "no"
			 forecast_income(no) ///
			 usd_year(`usd_year') /// USD year of income
			 inc_year(`=`year_move'+(`parent_earn_years'/2)') /// year of income measurement
			 earnings_type(household) ///
			 program_age(`=round(`age_stat',1)') // age we're projecting from
			 kids(`=round(`kids_per_family')')
			 
	local tax_rate_parent = r(tax_rate)
}

*Tax Impacts
if "`proj_type'" == "growth forecast" {

	local tax_impact_y = `tax_rate_young_short' * `tot_earn_impact_y'
	local tot_earn_impact_aftertax_y =  `tot_earn_impact_y' - `tax_impact_y'
	
	local tax_impact_y = `tax_impact_y' + (`tax_rate_young_long' * `tot_earn_impact_y')
	local tot_earn_impact_aftertax_y =  `tot_earn_impact_aftertax_y' + `tot_earn_impact_dy'*((1/(1+`discount_rate'))^(27-`avg_age_move_young_int')) - (`tax_rate_young_long' * `tot_earn_impact_y')
		
	if "`kids_age'" == "observed" | "`kids_age'" == "old"  {
	
		local tax_impact_o = (`tax_rate_old_short'*`tot_earn_impact_o')
		local tot_earn_impact_aftertax_o = `tot_earn_impact_do' * ((1/(1+`discount_rate'))^(18-`avg_age_move_old_int')) - (`tax_rate_old_short' * `tot_earn_impact_o')
		
		local tax_impact_o = `tax_impact_o' + `tax_rate_old_long' * `tot_earn_impact_o'
		local tot_earn_impact_aftertax_o = `tot_earn_impact_aftertax_o' + `tot_earn_impact_do'*((1/(1+`discount_rate'))^(27-`avg_age_move_old_int')) - (`tax_rate_old_long' * `tot_earn_impact_o')
	
	}
}

*Scale Effects
if "`proj_type'" != "no kids" {
	if "`kids_age'" == "young"{
		local tot_earn_impact = `tot_earn_impact_y'
		local tax_impact = `tax_impact_y'
		local tot_earn_impact_aftertax = `tot_earn_impact_aftertax_y'
	}

	if "`kids_age'" == "old"{
		local tot_earn_impact = `tot_earn_impact_o'
		local tax_impact = `tax_impact_o'
		local tot_earn_impact_aftertax = `tot_earn_impact_aftertax_o'
	}

	if "`kids_age'" == "observed" {
		local tot_earn_impact = (`tot_earn_impact_o'*(1 - `frac_kids_young')) + (`tot_earn_impact_y'*`frac_kids_young')
		local tax_impact = (`tax_impact_o'*(1 - `frac_kids_young')) + (`tax_impact_y'*`frac_kids_young')
		local tot_earn_impact_aftertax = (`tot_earn_impact_aftertax_o'*(1 - `frac_kids_young')) + (`tot_earn_impact_aftertax_y'*`frac_kids_young') 
	}
}
else if "`proj_type'" == "no kids" {
	local tot_earn_impact = 0
	local tax_impact = 0
	local tot_earn_impact_aftertax = 0
}

*Get parent impact
if "`include_parents'"=="yes" {

	local par_tot_earn_impact = 0
	
	if "`extend_parent_earn'"=="yes" local years = 65-round(`age_stat',1)
	else if "`extend_parent_earn'"=="no" local years = `parent_earn_years'
	
	forval i =1/`years' {
		local par_tot_earn_impact = `par_tot_earn_impact' + (`earn_impact_1_`parent_earn_years''/((1+`discount_rate')^(`i'-1)))
	}
	local FE_parent = `par_tot_welfare_impact'-`tax_rate_parent'*`par_tot_earn_impact'
}
else if "`include_parents'"=="no" local FE_parent = 0


*Get implied cost increase to govt from college attendance
if "$got_mto_cost"!="yes" {
	cost_of_college , year(`=`year_move_mto'+18-`avg_age_move_young_int'')
	global mto_college_cost_young = r(cost_of_college)
	global mto_tuition_cost_young = r(tuition)
	
	cost_of_college , year(`=`year_move_mto'+18-`avg_age_move_old_int'')
	global mto_college_cost_old = r(cost_of_college)
	global mto_tuition_cost_old = r(tuition)
	global got_mto_cost = "yes"
}

deflate_to `usd_year', from(`=`year_move_mto'+18-`avg_age_move_young_int'')
local govt_college_cost_young = (${mto_college_cost_young} - ${mto_tuition_cost_young})*`years_enroll'*(`college_effect_young'/100)* ///
	r(deflator)*(1/(1+`discount_rate')^(18-`avg_age_move_young_int'))

local priv_college_cost_young = (${mto_tuition_cost_young})*`years_enroll'*(`college_effect_young'/100)* ///
	r(deflator)*(1/(1+`discount_rate')^(18-`avg_age_move_young_int'))
	
deflate_to `usd_year', from(`=`year_move_mto'+18-`avg_age_move_old_int'')
local govt_college_cost_old = (${mto_college_cost_old} - ${mto_tuition_cost_old})*`years_enroll'*(`college_effect_old'/100)* ///
	r(deflator)*(1/(1+`discount_rate')^(18-`avg_age_move_old_int'))
local priv_college_cost_old = (${mto_tuition_cost_old})*`years_enroll'*(`college_effect_old'/100)* ///
	r(deflator)*(1/(1+`discount_rate')^(18-`avg_age_move_old_int'))

	
if "`kids_age'" == "old" {
	local govt_college_cost = `govt_college_cost_old' 
	local priv_college_cost = `priv_college_cost_old'
}

if "`kids_age'" == "young" {
	local govt_college_cost = `govt_college_cost_young' 
	local priv_college_cost = `priv_college_cost_young'
}

if "`kids_age'" == "observed" {
	local govt_college_cost = (`govt_college_cost_young'*`frac_kids_young') +  (`govt_college_cost_old'*(1 - `frac_kids_young'))
	local priv_college_cost = (`priv_college_cost_young'*`frac_kids_young') +  (`priv_college_cost_old'*(1 - `frac_kids_young'))
}

**************************
/* 5. Cost Calculations */
**************************

* Compute years of treatment for one "cycle" 
local years_treated = 18 - round(`avg_age_move_young',1)

local year_cost_abs = abs(`year_cost')

forval i = 1/`years_treated' {
	local program_cost_abs  = `program_cost_abs' + `year_cost_abs' * (1/(1+`discount_rate')^(`i'-1))
}
local program_cost = sign(`year_cost') * `program_cost_abs'


if "`include_kids'" == "yes" local FE = `FE_parent' - `tax_impact'*`kids_per_family' + `govt_college_cost'*`kids_per_family'
if "`include_kids'" == "no" local FE = `FE_parent'
di `FE_parent'+`program_cost'
di "`include_kids'"
local total_cost = `program_cost' + `FE' 


di `tax_impact'*`kids_per_family'
di `FE_parent'
di "`par_tot_welfare_impact'-`tax_rate_parent'*`par_tot_earn_impact'"
di `program_cost'  + `FE_parent' 
di `govt_college_cost'
di `tax_rate_parent' -0.3
di `total_cost'

*************************
/* 6. WTP Calculations */
*************************

di `tot_earn_impact'
di `FE_parent'
di  `tax_impact'
di `govt_college_cost'* `kids_per_family'
di `priv_college_cost'* `kids_per_family'  

if "`wtp_valuation'" == "post tax" {

	forval i = 1/`years_treated' {
		local WTP_adult = `WTP_adult' + `ev_coeff'*`year_val' * (1/(1+`discount_rate')^(`i'-1))
	}
	
	local WTP_kid = (`tot_earn_impact_aftertax' - `priv_college_cost')* `kids_per_family'  
	
	if "`proj_type'" == "no kids" {
		local WTP_kid = 0
	}
	
}

if "`wtp_valuation'" == "cost" {

	forval i = 1/`years_treated' {
		local WTP_adult = `WTP_adult' + `ev_coeff'*`year_val' * (1/(1+`discount_rate')^(`i'-1))
	}
	
	local WTP_kid = 0
}

if "`wtp_valuation'" == "lower bound" {

	forval i = 1/`years_treated' {
		local WTP_adult = `WTP_adult' + `ev_coeff'*`year_val' * (1/(1+`discount_rate')^(`i'-1))
	}
	local WTP_kid = 0
}

local WTP = `WTP_adult' + `WTP_kid'

**************************
/* 7. MVPF Calculations */
**************************

local MVPF = `WTP'/`total_cost'

****************
/* 8. Outputs */
****************

di `tot_earn_impact_y'
di `tot_earn_impact_o'
di `tot_earn_impact_aftertax'
di `tax_impact'
di `govt_college_cost_old'
di `govt_college_cost_young'
di `govt_college_cost'*`kids_per_family'
di `program_cost'
di `total_cost'
di `priv_college_cost'
di `WTP'
di `MVPF'

global program_cost_`1' = `program_cost'
global cost_`1' = `total_cost'
global WTP_`1' = `WTP'
global MVPF_`1' = `MVPF'

global age_stat_`1' = `age_stat'
if `WTP_kid'>`=`WTP'-`WTP_kid'' {
	global age_benef_`1' = `age_kid'
	}
else {
	global age_benef_`1' = `age_stat'
}	

* income globals
deflate_to 2015, from(`usd_year')
global inc_stat_`1' = `earn_control_1_10'*r(deflator)
global inc_type_stat_`1' = "household"
global inc_year_stat_`1' = `year_move'+(1+10)/2 // 1-10 years after move average
global inc_age_stat_`1' = `age_stat'+(1+10)/2

local cfactual_income_26 = (`MTO_control_earn_26_young'*`frac_kids_young') + (`MTO_control_earn_26_old'*(1 - `frac_kids_young'))
local avg_age_move_int = (`avg_age_move_young_int'*`frac_kids_young') + (`avg_age_move_old_int'*(1 - `frac_kids_young'))

if `WTP_kid'>`=0.5*`WTP'' {
	global inc_benef_`1' = `cfactual_income_26'*r(deflator)
	global inc_type_benef_`1' = "individual"
	global inc_year_benef_`1' = `year_move'+26-`avg_age_move_int'
	global inc_age_benef_`1' = 26
}
else {
	global inc_benef_`1' = `earn_control_1_10'*r(deflator)
	global inc_type_benef_`1' = "household"
	global inc_year_benef_`1' = `year_move'+(1+10)/2
	global inc_age_benef_`1' = `age_stat'+(1+10)/2
}

